GEOMETRIC TOOLS FOR COMPUTER GRAPHICS (MIRI)

Comparing Foley - Van Dam's and the quaternions methods: Rotating about the line through the origin with direction (1,1,1)

Vera Sacristán

Foley - Van Dam' s method

```
point = {x, y, z};
```

```
u = {1, 1, 1};
```

```
e = u / Sqrt[u.u]
```

$$\left\{\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right\}$$

```
MatrixForm[RotationOxBeta = {
   {1, 0, 0},
   {0, e[[3]], -e[[2]]} / Sqrt[e[[2]]^2 + e[[3]]^2],
   {0, e[[2]], e[[3]]} / Sqrt[e[[2]]^2 + e[[3]]^2]
  }]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} \\ 0 & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \end{pmatrix}$$

```
RotationOxBetaNeg = Transpose[RotationOxBeta];
```

```
ee = RotationOxBeta.e
```

$$\left\{\frac{1}{\sqrt{3}}, 0, \sqrt{\frac{2}{3}}\right\}$$

```
MatrixForm[RotationOyGamma = {
   {ee[[3]], 0, ee[[1]]} / Sqrt[ee[[1]]^2 + ee[[3]]^2],
   {0, 1, 0},
   {-ee[[1]], 0, ee[[3]]} / Sqrt[ee[[1]]^2 + ee[[3]]^2]
  }]
```

$$\begin{pmatrix} \sqrt{\frac{2}{3}} & 0 & \frac{1}{\sqrt{3}} \\ 0 & 1 & 0 \\ -\frac{1}{\sqrt{3}} & 0 & \sqrt{\frac{2}{3}} \end{pmatrix}$$

```
RotationOyGammaNeg = Transpose[RotationOyGamma];
```

```
MatrixForm[RotationOzAlpha = {
   {Cos[a], -Sin[a], 0},
   {Sin[a], Cos[a], 0},
   {0, 0, 1}
  }]
```

$$\begin{pmatrix} \text{Cos}[a] & -\text{Sin}[a] & 0 \\ \text{Sin}[a] & \text{Cos}[a] & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
RotationU = RotationOxBetaNeg.RotationOyGamma.
   RotationOzAlpha.RotationOyGammaNeg.RotationOxBeta;
```

```
rotatedPoint = Simplify[RotationU.point]
```

$$\left\{\frac{1}{3}\left(x+y+z+(2x-y-z)\,\text{Cos}[a]-\sqrt{3}\,(y-z)\,\text{Sin}[a]\right),\right.$$
$$\frac{1}{3}\left(x+y+z-(x-2y+z)\,\text{Cos}[a]+\sqrt{3}\,(x-z)\,\text{Sin}[a]\right),$$
$$\left.\frac{1}{3}\left(x+y+z-(x+y-2z)\,\text{Cos}[a]-\sqrt{3}\,(x-y)\,\text{Sin}[a]\right)\right\}$$

The quaternions method

```
Needs["Quaternions`"]
```

```
pointQ = Quaternion[0, x, y, z];
```

```
RotationQ = Quaternion[Cos[a / 2], Sin[a / 2] e[[1]],
   Sin[a / 2] e[[2]], Sin[a / 2] e[[3]]];
```

```
RotationQConj = Conjugate[RotationQ];
```

```
rotatedPointQ =
  Simplify[RotationQ ** pointQ ** RotationQConj]
```

$$\text{Quaternion}\left[0,\ \frac{1}{3}\left(x+y+z+(2x-y-z)\,\text{Cos}[a]-\sqrt{3}\,(y-z)\,\text{Sin}[a]\right),\right.$$
$$\frac{1}{3}\left(x+y+z-(x-2y+z)\,\text{Cos}[a]+\sqrt{3}\,(x-z)\,\text{Sin}[a]\right),$$
$$\left.\frac{1}{3}\left(x+y+z-(x+y-2z)\,\text{Cos}[a]-\sqrt{3}\,(x-y)\,\text{Sin}[a]\right)\right]$$

Comparing the two results

```
rotatedPointQ[[1]]
```

0

```
Table[rotatedPointQ[[i]], {i, 2, 4}]
```

$$\left\{\frac{1}{3}\left(x+y+z+(2x-y-z)\,\text{Cos}[a]-\sqrt{3}\,(y-z)\,\text{Sin}[a]\right),\ \frac{1}{3}\left(x+y+z-(x-2y+z)\,\text{Cos}[a]+\sqrt{3}\,(x-z)\,\text{Sin}[a]\right),\ \frac{1}{3}\left(x+y+z-(x+y-2z)\,\text{Cos}[a]-\sqrt{3}\,(x-y)\,\text{Sin}[a]\right)\right\}$$

```
rotatedPoint
```

$$\left\{\frac{1}{3}\left(x+y+z+(2x-y-z)\,\text{Cos}[a]-\sqrt{3}\,(y-z)\,\text{Sin}[a]\right),\ \frac{1}{3}\left(x+y+z-(x-2y+z)\,\text{Cos}[a]+\sqrt{3}\,(x-z)\,\text{Sin}[a]\right),\ \frac{1}{3}\left(x+y+z-(x+y-2z)\,\text{Cos}[a]-\sqrt{3}\,(x-y)\,\text{Sin}[a]\right)\right\}$$